Dataplot Vol 2 Vol 1

# KAPCDF

Name:
KAPCDF (LET)
Type:
Library Function
Purpose:
Compute the kappa cumulative distribution function with shape parameters h and k.
Description:
The general form of the kappa distribution has the following probability density function:

$$F(x;k,h,\xi,\alpha) = \left( 1 - h \left( 1 - \frac{k(x - \xi)}{\alpha} \right) ^{1/k} \right) ^{1/h} \hspace{20pt} \alpha > 0$$

with k and h denoting the shape parameters and $$\xi$$ and $$\alpha$$ denoting the location and scale parameters, respectively, and where F is the kappa cumulative distribution function.

The upper bound of x is

$$\begin{array}{ll} x < \xi + \alpha (1 - h^{-k}) & \mbox{ if } k > 0 \\ x < \infty & \mbox{ if } k \le 0 \end{array}$$

The lower bound of x is

$$\begin{array}{ll} x > \xi + \frac{\alpha (1 - h^{-k})}{k} & \mbox{ if } h > 0 \\ x > \frac{\xi \alpha}{k} & \mbox{ if } h \le 0, k < 0 \\ x > -\infty & \mbox{ if } h \le 0, k \ge 0 \end{array}$$

The standard form of the distribution is defined as $$\xi$$ = 0 and $$\alpha$$ = 1.

Syntax:
LET <y> = KAPCDF(<x>,<k>,<h>,<xi>,<alpha>)
<SUBSET/EXCEPT/FOR qualification>
where <x> is a number, parameter, or variable;
<k> is a number, parameter, or variable that specifies the first shape parameter;
<h> is a number, parameter, or variable that specifies the second shape parameter;
<xi> is a number, parameter, or variable that specifies the location parameter;
<alpha> is a number, parameter, or variable that specifies the scale parameter;
<y> is a variable or a parameter (depending on what <x> is) where the computed kappa cdf value is stored;
and where the <SUBSET/EXCEPT/FOR qualification> is optional.

The <xi> and <alpha> parameters are optional.

Examples:
LET A = KAPCDF(3,0.5,2,0,1.5)
LET X2 = KAPCDF(X1,K,H)
Note:
Dataplot uses Hoskings CDFKAP routine to compute the kappa cumulative distribution function. Hoskings report and associated Fortran code can be downloaded from the Statlib archive at

Default:
None
Synonyms:
None
Related Commands:
 KAPPDF = Compute the kappa probability density function. KAPPPF = Compute the kappa percent point function. MIEPDF = Compute Miekle's beta-kappa probability density function. GEVPDF = Compute the generalized extreme value probability density function. GEPPDF = Compute the generalized Pareto probability density function. GL5PDF = Compute the Hosking's generalized logistic probability density function.
References:
Hosking and Wallis (1997), "Regional Frequency Analysis", Cambridge University Press, Appendix A10.

J. R. M. Hosking (2000), "Research Report: Fortran Routines for use with the Method of L-Moments", IBM Research Division, T. J. Watson Research Center, Yorktown Heights, NY 10598.

Hoskings (1990), "L-moments: Analysis and Estimation of Distribution using Linear Combinations of Order Statistics", Journal of the Royal Statistical Society, Series B, 52, pp. 105-124.

Applications:
Distributional Modeling
Implementation Date:
2008/5
Program 1:

LET KP = DATA -0.5  0.1  0.5  1.0
LET H1 = -0.5
LET H2 = 0.1
LET H3 = 1
LET H4 = 2
.
MULTIPLOT 2 2
MULTIPLOT CORNER COORDINATES 0 0 95 95
MULTIPLOT SCALE FACTOR 2
TITLE CASE ASIS
TITLE OFFSET 2
X3LABEL
LINE COLOR BLACK RED BLUE GREEN
.
LOOP FOR KK = 1 1 4
LET K = KP(KK)
LET LL1 = KAPPPF(0.05,K,H1)
LET UL1 = KAPPPF(0.95,K,H1)
LET LL2 = KAPPPF(0.05,K,H2)
LET UL2 = KAPPPF(0.95,K,H2)
LET LL3 = KAPPPF(0.05,K,H3)
LET UL3 = KAPPPF(0.95,K,H3)
LET LL4 = KAPPPF(0.05,K,H4)
LET UL4 = KAPPPF(0.95,K,H4)
TITLE K = ^K, H = -0.5, 0.1, 1, 2
PLOT KAPCDF(X,K,H1) FOR X = LL1  0.01  UL1  AND
PLOT KAPCDF(X,K,H2) FOR X = LL2  0.01  UL2  AND
PLOT KAPCDF(X,K,H3) FOR X = LL3  0.01  UL3  AND
PLOT KAPCDF(X,K,H4) FOR X = LL4  0.01  UL4
END OF LOOP
END OF MULTIPLOT
.
JUSTIFICATION CENTER
MOVE 50 97
TEXT Kappa CDF Functions


NIST is an agency of the U.S. Commerce Department.

Date created: 07/07/2009
Last updated: 10/07/2016